# Equações Diferenciais EDI

## Método dos Coeficientes Indeterminados

2011/1

Prof. Fernando Deeke Sasse
CCT - UDESC

1. Considere a equação

$$m\left(\frac{d^2}{dt^2}x(t)\right) = -\kappa x + F_0 \cos(\omega t)\ .$$

Determine a solução geral, usando o método dos coeficientes indeterminados, quando (i) $\omega \neq \sqrt{\frac{\kappa}{m}}$ e (ii) $\omega = \sqrt{\frac{\kappa}{m}}$ .

Solução

(i)

```
> restart:
> eq:=m*(diff(x(t), t, t)) = -kappa*x(t)+F[0]*cos(omega*t);
```

$$eq := m\left(\frac{d^2}{dt^2}x(t)\right) = -\kappa\, x(t) + F_0 \cos(\omega t) \quad (1)$$

```
> dsolve(eq, x(t))
```

$$x(t) = \sin\left(\frac{\sqrt{\kappa}\, t}{\sqrt{m}}\right) \_C2 + \cos\left(\frac{\sqrt{\kappa}\, t}{\sqrt{m}}\right) \_C1 - \frac{F_0 \cos(\omega t)}{-\kappa + \omega^2 m} \quad (2)$$

(ii)

```
> omega:=sqrt(kappa/m):
> dsolve(eq, x(t));
```

$$x(t) = \sin\left(\frac{\sqrt{\kappa}\, t}{\sqrt{m}}\right) \_C2 + \cos\left(\frac{\sqrt{\kappa}\, t}{\sqrt{m}}\right) \_C1 + \frac{1}{4}\, \frac{F_0\left(\cos\left(\frac{\sqrt{\kappa}\, t}{\sqrt{m}}\right)\sqrt{\kappa}\sqrt{m} + 2 \sin\left(\frac{\sqrt{\kappa}\, t}{\sqrt{m}}\right) t\, \kappa\right)}{\kappa^{3/2}\sqrt{m}} \quad (3)$$

```
> expand(%);
```

$$x(t) = \sin\left(\frac{\sqrt{\kappa}\, t}{\sqrt{m}}\right) \_C2 + \cos\left(\frac{\sqrt{\kappa}\, t}{\sqrt{m}}\right) \_C1 + \frac{1}{4}\, \frac{F_0 \cos\left(\frac{\sqrt{\kappa}\, t}{\sqrt{m}}\right)}{\kappa} + \frac{1}{2}\, \frac{F_0 \sin\left(\frac{\sqrt{\kappa}\, t}{\sqrt{m}}\right) t}{\sqrt{\kappa}\sqrt{m}} \quad (4)$$

Note que o termo $\frac{1}{4}\,\frac{F_0 \cos\left(\frac{\sqrt{\kappa}\,t}{\sqrt{m}}\right)}{\kappa}$ pode ser absorvido pelo termo $\sin\left(\frac{\sqrt{\kappa}\,t}{\sqrt{m}}\right)$ _C2.

2. Resolva o problema de valor inicial

$$x^2 y'' - 3\,x y' + 3\,y = 2\,x^4 e^x, \quad y'(1) = 0, \; y(1) = 1.$$

```
> restart;
> eq:=x^2*diff(y(x),x$2)-3*x*diff(y(x),x)+3*y(x)=2*x^4*exp(x);
```

$$eq := x^2 \left(\frac{d^2}{dx^2}\,y(x)\right) - 3\,x\left(\frac{d}{dx}\,y(x)\right) + 3\,y(x) = 2\,x^4\,e^x \qquad (5)$$

```
> dsolve({eq, y(1) = 1, (D(y))(1) = 0}, y(x));
```

$$y(x) = \left(2\,x\,e^x - 2\,e^x + \frac{1}{2}\,x^2\,(-1 - 2\,e) + e + \frac{3}{2}\right)x \qquad (6)$$

3. Utilize o método dos coeficientes indeterminados para resolver a equação

$$y'' - 3\,y' - 4\,y = e^{-t}(5\,t - 1).$$

```
> restart;
> eq := diff(y(t), `$`(t, 2))-3*(diff(y(t), t))-4*y(t) = exp(-t)*
  (5*t-1);
```

$$eq := \frac{d^2}{dt^2}\,y(t) - 3\left(\frac{d}{dt}\,y(t)\right) - 4\,y(t) = e^{-t}\,(5\,t - 1) \qquad (7)$$

```
> dsolve(eq, y(t));
```

$$y(t) = e^{-t}\,\_C2 + e^{4\,t}\,\_C1 - \frac{1}{2}\,t^2\,e^{-t} \qquad (8)$$

Façamos o procedimento manual do método dos coeficientes indeterminados. A solução geral $y_h$ da parte homogênea é

```
> eq1:= s^2-3*s-4=0;
```

$$eq1 := s^2 - 3\,s - 4 = 0 \qquad (9)$$

```
> lambda:=solve(eq1,s);
```

$$\lambda := 4, -1 \qquad (10)$$

```
> yh:=C1*exp(lambda[1]*t)+C2*exp(lambda[2]*t);
```

$$yh := C1\,e^{4\,t} + C2\,e^{-t} \qquad (11)$$

A solução particular da parte não homogenea é dada por

```
> yp:=(A*t+B)*exp(-t);
```

$$yp := (A\,t + B)\,e^{-t} \qquad (12)$$

Como $B e^{-t}$ já é solução da parte homogênea, devemos usar:

```
> yp := t*(A*t+B)*exp(-t);
```

(13)

$$yp := t\,(A\,t + B)\,e^{-t} \qquad (13)$$

```
> eq2 := diff(yp, t, t)-3*(diff(yp, t))-4*yp - exp(-t)*(5*t-1);
```

$$eq2 := 2\,A\,e^{-t} - 5\,(A\,t + B)\,e^{-t} - 5\,t\,A\,e^{-t} - e^{-t}\,(5\,t - 1) \qquad (14)$$

```
> eq3:=simplify(eq2/exp(-t));
```

$$eq3 := 2\,A - 10\,A\,t - 5\,B - 5\,t + 1 \qquad (15)$$

```
> collect(eq3,t);
```

$$(-10\,A - 5)\,t + 2\,A - 5\,B + 1 \qquad (16)$$

```
> ee1:=coeff(eq3,t)=0;
```

$$ee1 := -10\,A - 5 = 0 \qquad (17)$$

```
> ee2:=coeff(eq3,t,0)=0;
```

$$ee2 := 2\,A - 5\,B + 1 = 0 \qquad (18)$$

```
> ss:=solve([ee1,ee2]);
```

$$ss := \left\{A = -\frac{1}{2},\, B = 0\right\} \qquad (19)$$

```
> assign(ss);
```

Portanto,

```
> yp;
```

$$-\frac{1}{2}\,t^2\,e^{-t} \qquad (20)$$

```
> y:=yh+yp;
```

$$y := C1\,e^{4\,t} + C2\,e^{-t} - \frac{1}{2}\,t^2\,e^{-t} \qquad (21)$$